# Dynamic stochastic general equilibrium models in Stata 15

David Schenck

Senior Econometrician
Stata

October 11, 2017

# Motivation

- Models used in macroeconomics for policy analysis

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy
- You need a model

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy
- You need a model
    - Dynamic

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy
- You need a model
  - Dynamic
  - Stochastic

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy
- You need a model
  - Dynamic
  - Stochastic
  - General equilibrium

# Motivation

- Models used in macroeconomics for policy analysis
- Suppose you want to understand the effect of some policy
- You need a model
    - Dynamic
    - Stochastic
    - General equilibrium
- and you need to estimate the model parameters

# Here's a model in words

- Consumers
  - Consume and save output
  - Take inflation and interest rate as given

# Here's a model in words

- Consumers
  - Consume and save output
  - Take inflation and interest rate as given
- Producers
  - Produce output and set prices
  - Take demand as given

# Here's a model in words

- Consumers
  - Consume and save output
  - Take inflation and interest rate as given
- Producers
  - Produce output and set prices
  - Take demand as given
- Central bank
  - Sets interest rate
  - Adjusts interest rate in response to inflation

# Here's a model in words

- Consumers
  - Consume and save output
  - Take inflation and interest rate as given
- Producers
  - Produce output and set prices
  - Take demand as given
- Central bank
  - Sets interest rate
  - Adjusts interest rate in response to inflation
- In equilibrium, this is a model that simultaneously determines output, inflation, and the interest rate

# Here's a model in equations I

- Consumers demand output ($x_t$), given inflation ($\pi_t$), the interest rate ($r_t$), and latent factors ($z_t$):

$$x_t = E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t)$$

# Here's a model in equations I

- Consumers demand output ($x_t$), given inflation ($\pi_t$), the interest rate ($r_t$), and latent factors ($z_t$):

$$x_t = E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t)$$

- Producers set prices, given output demand:

$$\pi_t = \beta E_t(\pi_{t+1}) + \kappa x_t$$

# Here's a model in equations I

- Consumers demand output ($x_t$), given inflation ($\pi_t$), the interest rate ($r_t$), and latent factors ($z_t$):

$$x_t = E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t)$$

- Producers set prices, given output demand:

$$\pi_t = \beta E_t(\pi_{t+1}) + \kappa x_t$$

- Central bank sets interest rate, given inflation

$$r_t = \frac{1}{\beta}\pi_t + u_t$$

# Here's a model in equations II

- The model's endogenous variables are characterized by equations:

$$
\begin{aligned}
x_t &= E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t) \\
\pi_t &= \beta E_t(\pi_{t+1}) + \kappa x_t \\
r_t &= \frac{1}{\beta}\pi_t + u_t
\end{aligned}
$$

# Here's a model in equations II

- The model's endogenous variables are characterized by equations:

$$x_t = E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t)$$
$$\pi_t = \beta E_t(\pi_{t+1}) + \kappa x_t$$
$$r_t = \frac{1}{\beta}\pi_t + u_t$$

- The model is completed by adding equations for the state variables:

$$z_{t+1} = \rho_z z_t + \xi_{t+1}$$
$$u_{t+1} = \rho_u u_t + \varepsilon_{t+1}$$

# Here's a model in Stata

- The model equations:

$$
\begin{aligned}
x_t &= E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t) \\
\pi_t &= \beta E_t(\pi_{t+1}) + \kappa x_t \\
r_t &= \frac{1}{\beta}\pi_t + u_t \\
z_{t+1} &= \rho_z z_t + \xi_{t+1} \\
u_{t+1} &= \rho_u u_t + \varepsilon_{t+1}
\end{aligned}
$$

# Here's a model in Stata

- The model equations:

$$
\begin{aligned}
x_t &= E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t) \\
\pi_t &= \beta E_t(\pi_{t+1}) + \kappa x_t \\
r_t &= \frac{1}{\beta}\pi_t + u_t \\
z_{t+1} &= \rho_z z_t + \xi_{t+1} \\
u_{t+1} &= \rho_u u_t + \varepsilon_{t+1}
\end{aligned}
$$

- In Stata:

```
. dsge    (x = E(F.x) - (r - E(F.p) - z), unobserved) ///
          (p = {beta}*E(F.p) + {kappa}*x)              ///
          (r = 1/{beta}*p + u)                         ///
          (F.z = {rhoz}*z, state)                      ///
          (F.u = {rhou}*u, state)
```

# Data and data management

- We will use data on inflation and interest rates
- Available from the Federal Reserve Economic Database (FRED) repository

```
. import fred GDPDEF FEDFUNDS, aggregate(quarterly, eop)
. generate dateq = qofd(daten)
. tsset dateq, quarterly

. generate lprice = ln(GDPDEF)
. generate p = 400*D.lprice

. rename FEDFUNDS r
```

# Data figure



20
15
10
5
0
1955q3
1970q3
1985q3
2000q3
2015q3
Date (quarters)
Growth rate of prices (GDPDEF)
Federal funds rate (FEDFUNDS)

# Parameter estimation

```
. dsge    (x = E(F.x) - (r - E(F.p) - z), unobserved) ///
>         (p = {beta}*E(F.p) + {kappa}*x)             ///
>         (r = 1/{beta}*p + u)                        ///
>         (F.z = {rhoz}*z, state)                     ///
>         (F.u = {rhou}*u, state), nolog
```

DSGE model

Sample: 1955q1 - 2015q4 Number of obs = 244
Log likelihood = -753.57131

| | Coef. | OIM Std. Err. | z | P>\|z\| | [95% Conf. | Interval] |
|---|---|---|---|---|---|---|
| /structural | | | | | | |
| beta | .5146675 | .0783489 | 6.57 | 0.000 | .3611065 | .6682284 |
| kappa | .1659054 | .0474072 | 3.50 | 0.000 | .0729889 | .2588218 |
| rhoz | .9545256 | .0186424 | 51.20 | 0.000 | .9179872 | .991064 |
| rhou | .7005486 | .0452604 | 15.48 | 0.000 | .6118398 | .7892573 |
| sd(e.z) | .6211211 | .101508 | | | .4221692 | .8200731 |
| sd(e.u) | 2.318202 | .3047436 | | | 1.720916 | 2.915489 |

# Policy questions

What is the effect of an unexpected increase in interest rates?

Estimated DSGE model provides an answer to this question. We can subject the model to a shock, then see how that shock feeds through the rest of the system.

# Impulse response setup

- First we set up the impulse response file:

```
. irf set dsge_irf
. irf create model1
```

# Impulse response setup

- First we set up the impulse response file:

```
. irf set dsge_irf
. irf create model1
```

- Then we draw the graph:

```
. irf graph irf, impulse(u) response(x p r u) byopts(yrescale) yline(0)
```

# Impulse responses from the estimated model



# Solving a DSGE Model

- Solution to a model is the key to estimation and generating impulse responses

# Solving a DSGE Model

- Solution to a model is the key to estimation and generating impulse responses
- Solution expresses endogenous variables as a function of state variables alone

# Viewing solution matrices

- You may wish to view the solution matrices
- Use the postestimation commands `estat policy` and `estat transition`

```
. estat policy
Policy matrix
```

| | | Coef. | Delta-method Std. Err. | z | P>\|z\| | [95% Conf. | Interval] |
|---|---|---|---|---|---|---|---|
| x | | | | | | | |
| | z | 2.718669 | .9420993 | 2.89 | 0.004 | .8721883 | 4.56515 |
| | u | -1.608225 | .4050569 | -3.97 | 0.000 | -2.402122 | -.8143284 |
| p | | | | | | | |
| | z | .8865884 | .2406229 | 3.68 | 0.000 | .4149761 | 1.358201 |
| | u | -.4172524 | .0393609 | -10.60 | 0.000 | -.4943985 | -.3401064 |
| r | | | | | | | |
| | z | 1.722641 | .2276535 | 7.57 | 0.000 | 1.276449 | 2.168834 |
| | u | .1892784 | .0591663 | 3.20 | 0.001 | .0733147 | .3052422 |

# Predictions of observed variables

- We can use the DSGE model to generate one-step-ahead predictions of the observed variables

```
. predict p_pred r_pred
(option xb assumed; fitted values)
```

# Prediction of inflation rate, $p$



# Prediction of interest rate, $r$



# Predictions of latent states

- We can use the DSGE model to generate estimates of the latent states

```
. predict z u, state
```

# Prediction of latent demand state, $z$



# Prediction of latent monetary state, $u$



# Using constraints to fix some parameters

You might want to fix some parameters and estimate others.

$$
\begin{aligned}
x_t &= E_t(x_{t+1}) - (r_t - E_t(\pi_{t+1}) - z_t) && \text{(Output demand)}\\
\pi_t &= \beta E_t(\pi_{t+1}) + \kappa x_t && \text{(Pricing)}\\
r_t &= \psi \pi_t + u_t && \text{(Interest rate)}\\
z_{t+1} &= \rho_z z_t + \xi_{t+1} && \text{(Demand shock)}\\
u_{t+1} &= \rho_u u_t + \varepsilon_{t+1} && \text{(Monetary shock)}
\end{aligned}
$$

- New: the $\psi$ parameter in the third equation.
- New: $\beta$ no longer appears in multiple equations.
- Assume: you wish to fix $\beta$ and estimate the remaining parameters conditional on your choice of $\beta$.

# Constrained model in Stata

```
. constraint 1 _b[beta]=0.96

. dsge     (x = E(F.x) - (r - E(F.p) - z), unobserved)   ///
           (p = {beta}*E(F.p) + {kappa}*x)               ///
           (r = {psi=1.5}*p + u)                         ///
           (F.z = {rhoz}*z, state)                       ///
           (F.u = {rhou}*u, state), constraint(1) nolog
```

# Parameter estimation with constraints

```
DSGE model

Sample: 1955q1 - 2015q4                         Number of obs     =        244
Log likelihood = -753.57131
 ( 1)  [/structural]beta = .96
------------------------------------------------------------------------------
             |                 OIM
             |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
/structural  |
        beta |        .96  (constrained)
       kappa |   .0849632   .0287693     2.95   0.003     .0285764    .1413501
         psi |   1.943004   .2957865     6.57   0.000     1.363273    2.522734
        rhoz |   .9545257   .0186424    51.20   0.000     .9179873     .991064
        rhou |   .7005482   .0452603    15.48   0.000     .6118396    .7892568
-------------+----------------------------------------------------------------
      sd(e.z)|    .568989   .0982973                      .3763299    .7616482
      sd(e.u)|   2.318204   .3047431                      1.720918    2.915489
------------------------------------------------------------------------------
```

# Conclusion

- `dsge` estimates the parameters of DSGE models
- Impulse response functions trace the effect of a shock on the model
- View the model's estimated reduced-form matrices
- Predictions of observable variables and latent states
- Constrained estimation

- Other features: solve a model at a particular parameter point (calibration), identification and stability diagnostics

Thank You!